# First SPICE Exercise

Fundamentals Of Electronics - a.a. 2018-2019 - University of Padua (Italy)

Pietro Prandini (mat. 1097752)

May 1, 2019



## 1 Audio amplifier

### 1.1 Voltage gain and frequency domain - Ideal op. amp.




Figure 1: Audio amplifier - Ideal op. amp.

By the analysis of the figure 1's circuit, It's possible to calculate the node $V_+$ voltage from the ratio of the voltage divider formed by $R_2$ and $C_2$, infact the node $V_+$ voltage is the same voltage of the resistance $R_2$ (equation 1).

$$V_+(s) = V_{in}(s)\frac{R_2}{R_2 + \frac{1}{sC_2}} = V_{in}(s)\frac{R_2}{R_2 + \frac{1}{sC_2}}\frac{sC_2}{sC_2} = V_{in}(s)\frac{sC_2R_2}{1 + sC_2R_2} \tag{1}$$

The negative feedback produces the virtual short circuit effect, so the $V_-$ and the $V_+$ voltages have virtually the same value (equation 2), and, because of the fact that the ideal operational amplifier $U_1$ isn't absorbe current from the $V_-$ and the $V_+$ nodes, the current of $I_{R_4}$ is the same current of $I_{R_3}$ (equation 3).
The current $I_{R_3}$ is calculated by the Ohm law (equation 4).

$$V_- = V_+ \tag{2}$$

$$I_{R_4} = I_{R_3} \tag{3}$$

$$I_{R_3} = \frac{V_-}{R_3} = \frac{V_+}{R_3} \tag{4}$$

By combining the past considerations it's possible to define the output voltage $V_o$ relating to the voltage input $V_{in}$ (equation 5).

$$V_o(s) = V_+(s) + R_4 I_{R_4} = V_+(s) + R_4 I_{R_3} = V_+(s) + R_4 \cdot \frac{V_+(s)}{R_3} = V_+(s) \cdot \left(1 + \frac{R_4}{R_3}\right) = V_{in}(s)\frac{sC_2R_2}{1 + sC_2R_2} \cdot \left(1 + \frac{R_4}{R_3}\right) \tag{5}$$

Conseguently of the equation 5, the transfer funtion $V_o(s)/V_{in}(s)$ is descripted by the equation 6.

$$\frac{V_o(s)}{V_{in}(s)} = \frac{sC_2R_2}{1+sC_2R_2}\left(1+\frac{R_4}{R_3}\right) \tag{6}$$

Defining $K$ as in the equation 7 and $\omega_1$ as in the equation 8, the transfer function $V_o(s)/V_{in}(s)$ became in the Bode form (equation 9).

$$K = C_2R_2\cdot\left(1+\frac{R_4}{R_3}\right) \tag{7}$$

$$\omega_1 = \frac{1}{C_2R_2} \tag{8}$$

$$\frac{V_o(s)}{V_{in}(s)} = K\frac{s}{1+s\frac{1}{\omega_1}} \tag{9}$$

Finally it's possible to calculate the frequency domain by the analysis of the transfer function's Bode form (equations 10 and 11).

$$K|_{dB} = 20\log_{10}|K| = \log_{10}\left|C_2R_2\cdot\left(1+\frac{R_4}{R_3}\right)\right| = -19.1722dB \tag{10}$$

$$\log_{10}|\omega_1| = \log_{10}\left|\frac{1}{C_2R_2}\right| = 2.0000 \tag{11}$$

## 1.2 Voltage output waveform - LT1028 op. amp.




Figure 2: Audio amplifier - LT1028 op. amp.

From now it's considered the circuit of the figure 2.

In order to simulate the waveform output voltage with a sinusoidal voltage input $V_{in}$ with an amplitude of $10mV$ and the frequencies of $1Hz$, $10Hz$ and $10kHz$, it's possible to use a SPICE transient analysis.

### 1.2.1 Netlist

It's presented the netlist for the SPICE analysis requested.

```
* Audio Amplifier - Waveform
***************************************************************************
* 1st Exercise - Fundamentals Of Electronics - a.a. 2018-2019 - UniPD - 1 of 4 *
*                    Pietro Prandini - mat. 1097752                         *
*                                                                           *
* This work is licensed under the Creative Commons Attribution-ShareAlike 4.0 *
* International License. To view a copy of this license, visit              *
```

```
* http://creativecommons.org/licenses/by-sa/4.0/ or send a letter to Creative *
* Commons, PO Box 1866, Mountain View, CA 94042, USA.                         *
********************************************************************************

* Libraries
.LIB LTC.lib

* Amplifiers
XU1 V+ V- A+ A- Vo LT1028

* Capacitances
C2 Vi V+ 100n

* Generators
Vin Vi 0 DC 0 AC 1 sin(0 10mV {F} 0 0 0)
Vdd A+ 0 DC 10
Vss A- 0 DC -10

* Resistances
R2 V+ 0 100k
R3 V- 0 1k
R4 Vo V- 10k

* Analysis
.step param F list 1Hz 10Hz 100Hz
.tran 0 250m 0 1m uic

.END
```

### 1.2.2 Graph

This graph is the output of the last netlist presented. There are three curves, one for every frequency analyzed ($1Hz$, $10Hz$ and $10kHz$).




Figure 3: Audio Amplifier - Voltage output waveform

## 1.3 Bode plot - LT1028 op. amp.

The Bode plot could be generated with a SPICE small signal AC analysis.

### 1.3.1 Netlist

It's presented the netlist for the SPICE analysis requested.

```
* Audio Amplifier - Bode diagram
*****************************************************************************
* 1st Exercise - Fundamentals Of Electronics - a.a. 2018-2019 - UniPD - 1 of 4 *
*                      Pietro Prandini - mat. 1097752                          *
*                                                                              *
* This work is licensed under the Creative Commons Attribution-ShareAlike 4.0  *
* International License. To view a copy of this license, visit                *
* http://creativecommons.org/licenses/by-sa/4.0/ or send a letter to Creative  *
* Commons, PO Box 1866, Mountain View, CA 94042, USA.                          *
*****************************************************************************

* Libraries
.LIB LTC.lib

* Amplifiers
XU1 V+ V- A+ A- Vo LT1028

* Capacitances
C2 Vi V+ 100n

* Generators
Vin Vi 0 DC 0 AC 1 sin(0 10mV {F} 0 0 0)
Vdd A+ 0 DC 10
Vss A- 0 DC -10

* Resistances
R2 V+ 0 100k
R3 V- 0 1k
R4 Vo V- 10k

* Analysis
.step param F list 1Hz 10Hz 100Hz
.ac DEC 10 1 100k

.END
```

### 1.3.2 Graph

The Bode plot generated could be visible in the figure 4. The continuous line represents the magnitude and the dashed line represents the phase frequency.




Figure 4: Audio Amplifier - Bode plot

### 1.4 Saturation - LT1028 op. amp.

The voltage output saturation could be analized by giving an abnormally high voltage input to the input. The next netlist analyzes the voltage output with an input with $100V$ amplitude.

```
* Audio Amplifier - Waveform with abnormal input voltage amplitude
*************************************************************************
* 1st Exercise - Fundamentals Of Electronics - a.a. 2018-2019 - UniPD - 1 of 4 *
*                          Pietro Prandini - mat. 1097752                       *
*                                                                               *
* This work is licensed under the Creative Commons Attribution-ShareAlike 4.0  *
* International License. To view a copy of this license, visit                 *
* http://creativecommons.org/licenses/by-sa/4.0/ or send a letter to Creative  *
* Commons, PO Box 1866, Mountain View, CA 94042, USA.                           *
*************************************************************************

* Libraries
.LIB LTC.lib

* Amplifiers
XU1 V+ V- A+ A- Vo LT1028

* Capacitances
C2 Vi V+ 100n

* Generators
Vin Vi 0 DC 0 AC 1 sin(0 100V 10Hz 0 0 0)
Vdd A+ 0 DC 10
Vss A- 0 DC -10

* Resistances
R2 V+ 0 100k
R3 V- 0 1k
R4 Vo V- 10k
```

```
* Analysis
.tran 0 250m 0 1m uic

.END
```




Figure 5: Audio Amplifier - Output voltage saturation

The graph generated (figure 5) makes clear the fact that the voltage output saturation is on $\pm 8V$. So it's possible to calculate wich is the highest input signal that avoids the saturation (equation 12).

$$\begin{aligned}
|V_o| &\leq 8V \\
\left| V_{in} \cdot \frac{j\omega C_2 R_2}{1 + j\omega C_2 R_2} \cdot \left(1 + \frac{R_4}{R_3}\right)\right| &\leq 8V \\
|V_{in}| \cdot \left|\frac{j\omega C_2 R_2}{1 + j\omega C_2 R_2}\right| \cdot \left|\left(1 + \frac{R_4}{R_3}\right)\right| &\leq 8V \\
|V_{in}| \cdot \frac{\sqrt{(\omega C_2 R_2)^2}}{\sqrt{1 + (\omega C_2 R_2)^2}} \cdot \frac{R_3 + R_4}{R_3} &\leq 8V \\
|V_{in}| \leq 8V \cdot \frac{\sqrt{1 + (\omega C_2 R_2)^2}}{\omega C_2 R_2} \cdot \frac{R_3}{R_3 + R_4} &
\end{aligned} \tag{12}$$

In order to have a maximum value of the input voltage $V_{in}$, the $\omega$ could be setted to a static value. For example with $\omega = 10Hz \cdot 2\pi$ the maximum value of the input voltage is $|V_{in}| \leq 1.36701V$ (output voltage graph on figure 6).




Figure 6: Audio Amplifier - Output voltage with an amplitude of the imput voltage equal to $1.36701V$, frequency equal to 10Hz

### 1.4.1 Output voltage waveform - $V_{in} = 2 \cdot |V_{inMAX}|$

The netlist used for plot the output voltage waveform with an input voltage dobled relatively to the maximum input voltage calculated is:

```
* Audio Amplifier - Waveform Vin = 2*VinMax
****************************************************************************
* 1st Exercise - Fundamentals Of Electronics - a.a. 2018-2019 - UniPD - 1 of 4 *
*                        Pietro Prandini - mat. 1097752                        *
*                                                                              *
* This work is licensed under the Creative Commons Attribution-ShareAlike 4.0  *
* International License. To view a copy of this license, visit                *
* http://creativecommons.org/licenses/by-sa/4.0/ or send a letter to Creative  *
* Commons, PO Box 1866, Mountain View, CA 94042, USA.                          *
****************************************************************************

* Libraries
.LIB LTC.lib

* Amplifiers
XU1 V+ V- A+ A- Vo LT1028

* Capacitances
C2 Vi V+ 100n

* Generators
Vin Vi 0 DC 0 AC 1 sin(0 {2*1.36701V} 10Hz 0 0 0)
Vdd A+ 0 DC 10
Vss A- 0 DC -10

* Resistances
R2 V+ 0 100k
R3 V- 0 1k
R4 Vo V- 10k
```

```
* Analysis
.tran 0 250m 0 1m uic

.END
```

The graph generated is presented on the figure 7.




Figure 7: Audio Amplifier - Output voltage with an amplitude of the imput voltage equal to $2 * 1.36701V$, frequency equal to 10Hz