

# Third SPICE Exercise

Fundamentals Of Electronics - a.a. 2018-2019 - University of Padua (Italy)

Pietro Prandini (mat. 1097752)

June 8, 2019



# Contents

<b>1</b>	<b>Differential amplifier with MOS current source</b>	<b>5</b>
1.1	MOSFET current mirror source - Analytic solution . . . . .	6
1.1.1	$V_{D_4SS}$ . . . . .	6
1.1.2	$I_{D_4}$ . . . . .	6
1.2	SPICE Operating Point analysis . . . . .	6



# Differential amplifier with MOS current source




Figure 1.1: Differential amplifier with MOS current source

Data:

$$V_t = 0.5V \quad (1.1)$$

$$K'_n = \mu_n C_{ox} = 200 \frac{\mu A}{V^2} \quad (1.2)$$

$$\lambda = 0 \quad (1.3)$$

$$\left(\frac{W}{L}\right)_1 = \left(\frac{W}{L}\right)_2 = 20 \quad (1.4)$$

$$\left(\frac{W}{L}\right)_3 = \left(\frac{W}{L}\right)_4 = 5 \quad (1.5)$$

$$R_{D_1} = R_{D_2} = 20k\Omega \quad (1.6)$$

$$R_{D_4} = \frac{30}{1000} \cdot 1097752\Omega = 32.93k\Omega \simeq 33k\Omega \quad (1.7)$$

$$V_{DD} = 3V \quad (1.8)$$

$$V_{SS} = -3V \quad (1.9)$$

## 1.1 MOSFET current mirror source - Analytic solution

### 1.1.1 $V_{D_4SS}$

The transistor  $M_4$  has a short circuit between its drain and its gate, so the transistor works in saturation mode and the voltage between the drain and the gate are the same of the voltage between the gate and the source. The current  $I_{D_4}$  could be calculated as:

$$I_{D_4} = \frac{1}{2} K'_n \left( \frac{W}{L} \right)_4 (V_{D_4SS} - V_t)^2 \quad (1.10)$$

Other expression of the current  $I_{D_4}$  could be calculated using the LKT:

$$V_{DD} - R_{D_4} I_{D_4} - V_{D_4SS} - V_{SS} = 0 \implies I_{D_4} = \frac{V_{DD} - V_{D_4SS} - V_{SS}}{R_{D_4}} \quad (1.11)$$

Using the equations 1.10 and 1.11 it's possible calculating  $V_{D_4SS}$ :

$$\frac{1}{2} K'_n \left( \frac{W}{L} \right)_4 (V_{D_4SS} - V_t)^2 = \frac{V_{DD} - V_{D_4SS} - V_{SS}}{R_{D_4}} \quad (1.12)$$

$$\frac{1}{2} \cdot 200 \frac{\mu A}{V^2} \cdot 5 \frac{\mu m}{\mu m} (V_{D_4SS} - 0.5V)^2 = \frac{3V - V_{D_4SS} - (-3V)}{33k\Omega} \quad (1.13)$$

$$500 \frac{\mu A}{V^2} (V_{D_4SS} - 0.5V)^2 = \frac{6}{33} mA - \frac{1}{33k\Omega} V_{D_4SS} \quad (1.14)$$

$$500 \frac{\mu A}{V^2} (V_{D_4SS}^2 - V_{D_4SS} \cdot V + 0.25V^2) = \frac{6}{33} mA - \frac{1}{33k\Omega} V_{D_4SS} \quad (1.15)$$

$$500 \frac{\mu A}{V^2} \cdot V_{D_4SS}^2 + \left( -500 \frac{\mu A}{V^2} V + \frac{1}{33k\Omega} \right) V_{D_4SS} + 500 \frac{\mu A}{V^2} \cdot 0.25V^2 - \frac{6}{33} mA = 0 \quad (1.16)$$

$$0.5 \frac{mA}{V^2} \cdot V_{D_4SS}^2 + \left( -0.5 \frac{mA}{V^2} V + \frac{1}{33k\Omega} \right) V_{D_4SS} + 0.5 \frac{mA}{V^2} \cdot 0.25V^2 - \frac{6}{33} mA = 0 \quad (1.17)$$

$$\left( 0.5 \frac{mA}{V^2} \right) V_{D_4SS}^2 + \left( -\frac{31}{66} \frac{mA}{V} \right) V_{D_4SS} + \left( -\frac{5}{88} mA \right) = 0 \quad (1.18)$$

$$V_{D_4SS1,2} = \frac{-\left(-\frac{31}{66} \frac{mA}{V}\right) \pm \sqrt{\left(-\frac{31}{66} \frac{mA}{V}\right)^2 - 4 \cdot \left(0.5 \frac{mA}{V^2}\right) \cdot \left(-\frac{5}{88} mA\right)}}{2 \cdot 0.5 \frac{mA}{V^2}} = \begin{cases} 1.04784V \\ -0.10845V \end{cases} \quad \begin{matrix} \text{Not possible: } < \text{ of } V_t \end{matrix} \quad (1.19)$$

### 1.1.2 $I_{D_4}$

Using the equation 1.10 and the result of the equation 1.19:

$$I_{D_4} = \frac{1}{2} \cdot 200 \mu A / V^2 \cdot 5 \frac{\mu m}{\mu m} \cdot (1.04784V - 0.5V)^2 = 150.06433 \mu A \quad (1.20)$$

## 1.2 SPICE Operating Point analysis

```
* Differential Amplifier
*****
* 3st Exercise – Fundamentals Of Electronics – a.a. 2018–2019 – UniPD – Italy *
*                               Pietro Prandini – mat. 1097752                               *
*                                                                                               *
* This work is licensed under the Creative Commons Attribution–ShareAlike 4.0 *
* International License. To view a copy of this license, visit                               *
* http://creativecommons.org/licenses/by-sa/4.0/ or send a letter to Creative *
* Commons, PO Box 1866, Mountain View, CA 94042, USA.                                     *
*****
```

```

* Parameters
.param RD = 20k
.param VG1 = 0
.param VG2 = 0

* NMOS models
.model NA NMOS VT0=0.5 KP=200u LAMBDA=0 W=4.00u L=0.20u
.model NS NMOS VT0=0.5 KP=200u LAMBDA=0 W=1.25u L=0.25u

* Resistances
RD1 DD D1 {RD}
RD2 DD D2 {RD}
RD4 DD D4 33k

* Transistors
M1 D1 G1 SA SA NA
M2 D2 G2 SA SA NA
M3 SA D4 SS SS NS
M4 D4 D4 SS SS NS

* Voltage sources
VDD DD 0 3
VSS SS 0 -3
Vi1 G1 0 {VG1}
Vi2 G2 0 {VG2}

* Analysis
.op

.END

```

—— Operating Point ——

V(dd):	3	voltage
V(d1):	1.49935	voltage
V(d2):	1.49935	voltage
V(d4):	-1.95216	voltage
V(g1):	0	voltage
V(sa):	-0.693691	voltage
V(g2):	0	voltage
V(ss):	-3	voltage
Id(M4):	0.000150065	device_current
Ig(M4):	0	device_current
Ib(M4):	-1.05784e-012	device_current
Is(M4):	-0.000150065	device_current
Id(M3):	0.000150065	device_current
Ig(M3):	0	device_current
Ib(M3):	-2.31631e-012	device_current
Is(M3):	-0.000150065	device_current
Id(M2):	7.50327e-005	device_current
Ig(M2):	0	device_current
Ib(M2):	-2.20304e-012	device_current
Is(M2):	-7.50327e-005	device_current
Id(M1):	7.50327e-005	device_current
Ig(M1):	0	device_current
Ib(M1):	-2.20304e-012	device_current
Is(M1):	-7.50327e-005	device_current
I(Rd4):	0.000150065	device_current
I(Rd2):	7.50327e-005	device_current
I(Rd1):	7.50327e-005	device_current

I(Vi2):	0	device_current
I(Vi1):	0	device_current
I(Vss):	0.000300131	device_current
I(Vdd):	-0.000300131	device_current