

Third SPICE Exercise

Fundamentals Of Electronics - a.a. 2018-2019 - University of Padua (Italy)

Pietro Prandini (mat. 1097752)

June 15, 2019

Contents

1	Differential amplifier with MOS current source	5
1.1	Initial data	5
1.2	Static conditions	6
1.2.1	MOSFET M_4	6
	Saturation mode checks	6
	V_{D_4SS} ($= V_{G_4SS}$)	6
	I_{D_4}	7
1.2.2	MOSFET M_3	7
	V_{G_3SS}	7
	I_{S_A}	7
	Saturation mode checks	7
1.2.3	MOSFET M_1 and MOSFET M_2	8
	I_{D_1} ($= I_{D_2}$)	8
	$V_{G_1S_A}$ ($= V_{G_2S_A}$)	8
	Saturation mode checks	8
1.2.4	MOSFET V_{DSQ} , V_{GSQ} , I_{DQ} - Resuming	9
1.2.5	g_m	9
1.3	Small signal analysis	9
1.3.1	MOSFET M_3 with $\lambda = 0.02$	9
1.3.2	Differential pure signal gain	9
1.3.3	Common mode gain	10
1.3.4	Common mode gain with $\lambda_3 = 0$	12
1.4	SPICE analysis	14
1.4.1	$R_D = 20k$	14
	Operating Point on static conditions ($\lambda_3 = 0$)	14
	Operating Point - common mode signals	15
	Operating Point - differential signals	16
1.4.2	$R_D = 1097752/100$	17
	Operating Point on static conditions ($\lambda_3 = 0$)	17
	Operating Point - common mode signals	19
	Operating Point - differential signals	20
1.4.3	$R_D = 1097752/1000$	21
	Operating Point on static conditions ($\lambda_3 = 0$)	21
	Operating Point - common mode signals	23
	Operating Point - differential signals	24
2	CMOS Inverter	27
2.1	Initial data	27
2.2	Dimensioning a balanced inverter	27
2.2.1	Resuming	28
2.3	Characteristic V_{IN}/V_{OUT}	28
2.4	Applying a square wave on input	29
2.5	Characteristic V_{IN}/V_{OUT} - Changing W_n	30
2.6	Applying a square wave on input - Changing W_n	31

Chapter 1

Differential amplifier with MOS current source




Figure 1.1: Differential amplifier with MOS current source

1.1 Initial data

$$V_t = 0.5V \quad (1.1)$$

$$K'_n = \mu_n C_{ox} = 200 \frac{\mu A}{V^2} \quad (1.2)$$

$$\lambda = 0 \quad (1.3)$$

$$\left(\frac{W}{L}\right)_1 = \left(\frac{W}{L}\right)_2 = 20 \quad (1.4)$$

$$\left(\frac{W}{L}\right)_3 = \left(\frac{W}{L}\right)_4 = 5 \quad (1.5)$$

$$R_{D1} = R_{D2} = R_D = 20k\Omega \quad (1.6)$$

$$R_{D4} = \frac{30}{1000} \cdot 1097752\Omega = 32.93k\Omega \simeq 33k\Omega \quad (1.7)$$

$$V_{DD} = 3V \quad (1.8)$$

$$V_{SS} = -3V \quad (1.9)$$

1.2 Static conditions




Figure 1.2: Differential amplifier with MOS current source - Static conditions

On static conditions it's considered the input signals V_{i1} and V_{i2} turned off. The equivalent circuit is on figure 1.2.

1.2.1 MOSFET M_4

Saturation mode checks

The transistor M_4 has a short circuit between its drain and its gate, so the transistor works in saturation mode and the voltage between the drain and the gate are the same of the voltage between the gate and the source:

$$V_{D_4SS} > V_{G_4SS} - V_t \xrightarrow{V_{G_4SS}=V_{D_4SS}} V_{D_4SS} > V_{D_4SS} - V_t \quad (1.10)$$

$$0 > -V_t \quad (\text{Always true: } V_t > 0) \quad (1.11)$$

It's requested to pay attention to the another check to confirm the work on saturation mode:

$$V_{G_4SS} > V_t \xrightarrow{V_{G_4SS}=V_{D_4SS}} V_{D_4SS} > V_t \quad (1.12)$$

$$V_{D_4SS} \quad (= V_{G_4SS})$$

Supposing that the transistor M_4 works on the saturation mode (see the section 1.2.1 for details), the current I_{D_4} could be calculated as:

$$I_{D_4} = \frac{1}{2} K'_n \left(\frac{W}{L} \right)_4 (V_{D_4SS} - V_t)^2 \quad (1.13)$$

Other expression of the current I_{D_4} could be calculated using the LKT:

$$V_{DD} - R_{D_4} I_{D_4} - V_{D_4SS} - V_{SS} = 0 \implies I_{D_4} = \frac{V_{DD} - V_{D_4SS} - V_{SS}}{R_{D_4}} \quad (1.14)$$

Using the equations 1.13 and 1.14 it's possible calculating V_{D_4SS} :

$$\frac{1}{2} K'_n \left(\frac{W}{L} \right)_4 (V_{D_4SS} - V_t)^2 = \frac{V_{DD} - V_{D_4SS} - V_{SS}}{R_{D_4}} \quad (1.15)$$

$$\frac{1}{2} \cdot 200 \frac{\mu A}{V^2} \cdot 5 \frac{\mu m}{\mu m} (V_{D_4SS} - 0.5V)^2 = \frac{3V - V_{D_4SS} - (-3V)}{33k\Omega} \quad (1.16)$$

$$500 \frac{\mu A}{V^2} (V_{D_4SS} - 0.5V)^2 = \frac{6}{33} mA - \frac{1}{33k\Omega} V_{D_4SS} \quad (1.17)$$

$$500 \frac{\mu A}{V^2} (V_{D_4SS}^2 - V_{D_4SS} \cdot V + 0.25V^2) = \frac{6}{33} mA - \frac{1}{33k\Omega} V_{D_4SS} \quad (1.18)$$

$$500 \frac{\mu A}{V^2} \cdot V_{D_4SS}^2 + \left(-500 \frac{\mu A}{V^2} V + \frac{1}{33k\Omega} \right) V_{D_4SS} + 500 \frac{\mu A}{V^2} \cdot 0.25V^2 - \frac{6}{33} mA = 0 \quad (1.19)$$

$$0.5 \frac{mA}{V^2} \cdot V_{D_4SS}^2 + \left(-0.5 \frac{mA}{V^2} V + \frac{1}{33k\Omega} \right) V_{D_4SS} + 0.5 \frac{mA}{V^2} \cdot 0.25V^2 - \frac{6}{33} mA = 0 \quad (1.20)$$

$$\left(0.5 \frac{mA}{V^2} \right) V_{D_4SS}^2 + \left(-\frac{31}{66} \frac{mA}{V} \right) V_{D_4SS} + \left(-\frac{5}{88} mA \right) = 0 \quad (1.21)$$

$$V_{D_4SS_{1,2}} = \frac{-\left(-\frac{31}{66} \frac{mA}{V}\right) \pm \sqrt{\left(-\frac{31}{66} \frac{mA}{V}\right)^2 - 4 \cdot \left(0.5 \frac{mA}{V^2}\right) \cdot \left(-\frac{5}{88} mA\right)}}{2 \cdot 0.5 \frac{mA}{V^2}} = \begin{cases} 1.04784V \\ -0.10845V \end{cases} \quad \text{Not possible: } < \text{ of } V_t \quad (1.22)$$

Now it's possible to check the last equation that can confirm the work on saturation mode of the MOSFET M_4 (equation 1.12):

$$1.04784V > 0.5V \quad M_4 \text{ works on saturation mode.} \quad (1.23)$$

I_{D_4}

Using the equation 1.13 and the result of the equation 1.22:

$$I_{D_4} = \frac{1}{2} \cdot 200 \mu A / V^2 \cdot 5 \frac{\mu m}{\mu m} \cdot (1.04784V - 0.5V)^2 = 150.06433 \mu A \quad (1.24)$$

1.2.2 MOSFET M_3

V_{G_3SS}

Observing the circuit represented on the figure 1.2 it's clear that the voltage V_{G_3SS} is equal to the voltage V_{D_4SS} calculated in the equation 1.22.

$$V_{G_3SS} = V_{D_4SS} \quad (1.25)$$

I_{S_A}

As agree with the consideration of the section 1.2.2 and supposing the work of the MOSFET M_3 on the saturation mode, it's possible calculating the drain current of the MOSFET M_3 :

$$I_{S_A} = \frac{1}{2} K'_n \left(\frac{W}{L} \right)_3 (V_{D_4SS} - V_t)^2 \quad (1.26)$$

$$I_{S_A} = \frac{1}{2} \cdot 200 \mu A / V^2 \cdot 5 \frac{\mu m}{\mu m} \cdot (1.04784V - 0.5V)^2 = 150.06433 \mu A \quad (1.27)$$

Saturation mode checks

For obtaining the confirm of the work of the MOSFET M_3 on saturation mode the next two equations have to be satisfied:

$$V_{D_3SS} > V_{G_3SS} - V_t \xrightarrow{V_{D_3SS}=V_{S_ASS}, V_{G_3SS}=V_{D_4SS}} V_{S_ASS} > V_{D_4SS} - V_t \quad (1.28)$$

$$V_{G_3SS} > V_t \xrightarrow{V_{G_3SS}=V_{D_4SS}} V_{D_4SS} > V_t \quad (1.29)$$

The equation 1.28 hasn't to be checked, V_{S_ASS} isn't calculated yet.
The equation 1.29 is satisfied (see the equation 1.23).

1.2.3 MOSFET M_1 and MOSFET M_2

The MOSFET M_1 and the MOSFET M_2 have the same dimension and the same constructive parameters (see initial data at the start of this chapter 1.1).

They also have the same voltage applied to every their pins (see figure 1.2).

So, additionally supposing the work on the saturation mode of M_1 and M_2 , it's possible to confirm the next equations:

$$I_{D_1} = I_{D_2} \quad (1.30)$$

$$V_{G_1 S_A} = V_{G_2 S_A} \quad (1.31)$$

$$V_{D_1 S_A} = V_{D_2 S_A} \quad (1.32)$$

$$I_{D_1} (= I_{D_2})$$

$$\text{LKC node } S_A: I_{D_1} + I_{D_2} - I_{S_A} = 0 \xrightarrow{\text{eq.1.30}} 2I_{D_1} - I_{S_A} = 0 \implies I_{D_1} = \frac{I_{S_A}}{2} \quad (1.33)$$

$$I_{D_1} = \frac{150.06433\mu A}{2} = 75.03217\mu A \quad (1.34)$$

$$V_{G_1 S_A} (= V_{G_2 S_A})$$

Supposing the work of the MOSFET M_1 (equally M_2) on the saturation mode, the drain current could be calculated as:

$$I_{D_1} = \frac{1}{2} K'_n \left(\frac{W}{L} \right)_1 (V_{G_1 S_A} - V_t)^2 \quad (1.35)$$

It's possible using the equation 1.35 to calculate $V_{G_1 S_A}$:

$$I_{D_1} = \frac{1}{2} K'_n \left(\frac{W}{L} \right)_1 (V_{G_1 S_A} - V_t)^2 \quad (1.36)$$

$$\sqrt{I_{D_1}} = \sqrt{\frac{1}{2} K'_n \left(\frac{W}{L} \right)_1} (V_{G_1 S_A} - V_t) \quad (1.37)$$

$$\sqrt{\frac{I_{D_1}}{\frac{1}{2} K'_n \left(\frac{W}{L} \right)_1}} = V_{G_1 S_A} - V_t \quad (1.38)$$

$$V_{G_1 S_A} = \sqrt{\frac{2I_{D_1}}{K'_n \left(\frac{W}{L} \right)_1}} + V_t \quad (1.39)$$

$$V_{G_1 S_A} = \sqrt{\frac{2 \cdot 75.03217\mu A}{200\mu A \cdot 20 \frac{\mu A}{\mu A}}} + 0.5V \quad (1.40)$$

$$V_{G_1 S_A} = 0.69369V \quad (1.41)$$

V_{S_A} Now it's possible calculating the voltage on the node S_A :

$$V_{G_1 S_A} = V_{G_1} - V_{S_A} \implies V_{S_A} = V_{G_1} - V_{G_1 S_A} \xrightarrow{V_{G_1}=0, \text{eq.1.41}} V_{S_A} = -0.69369V \quad (1.42)$$

Saturation mode checks

In order to check the mode of the M_1 (and M_2) the equations to respect are 1.43 and 1.48.

$$V_{D_1 S_A} > V_{G_1 S_A} - V_t \quad (1.43)$$

$$V_{D_1} - V_{S_A} > V_{G_1 S_A} - V_t \quad (1.44)$$

$$I_{D_1} R_{D_1} - V_{S_A} > V_{G_1 S_A} - V_t \quad (1.45)$$

$$75.03217\mu A \cdot 20k\Omega - (-0.69369V) > 0.69369V - 0.5V \quad (1.46)$$

$$2.19433V > 0.19369V \quad \text{True.} \quad (1.47)$$

$$V_{G_1S_A} > V_t \quad (1.48)$$

$$0.69369V > 0.5V \quad \text{True.} \quad (1.49)$$

Checking the mode of the M_3 's work (see equation 1.28):

$$V_{S_ASS} > V_{D_4SS} - V_t \quad (1.50)$$

$$V_{S_A} - V_{SS} > V_{D_4SS} - V_t \quad (1.51)$$

$$-0.69369V - (-3V) > 1.04784V - 0.5V \quad (1.52)$$

$$2.30631V > 0.54784V \quad M_3 \text{ works on the saturation mode.} \quad (1.53)$$

1.2.4 MOSFET V_{DS_Q} , V_{GS_Q} , I_{D_Q} - Resuming

MOSFET	V_{DS_Q}	V_{GS_Q}	I_{D_Q}
M_1	$V_{D_1S_A} = 2.19433V$	$V_{G_1S_A} = 0.69369V$	$I_{D_1} = 75.03217\mu A$
M_2	$V_{D_2S_A} = 2.19433V$	$V_{G_2S_A} = 0.69369V$	$I_{D_2} = 75.03217\mu A$
M_3	$V_{S_ASS} = 2.30631V$	$V_{D_4SS} = 1.04784V$	$I_{S_A} = 150.06433\mu A$
M_4	$V_{D_4SS} = 1.04784V$	$V_{D_4SS} = 1.04784V$	$I_{D_4} = 150.06433\mu A$

1.2.5 g_m

$$g_{m_1} = g_{m_2} = K'_n \left(\frac{W}{L} \right)_1 (V_{G_1S_A} - V_t) \quad (1.54)$$

$$= 200 \frac{\mu A}{V^2} \cdot 20 \frac{\mu A}{\mu A} \cdot (0.69369V - 0.5V) \quad (1.55)$$

$$= 774.76\mu A/V \quad (1.56)$$

1.3 Small signal analysis

1.3.1 MOSFET M_3 with $\lambda = 0.02$

From now the MOSFET M_3 is considered with $\lambda = 0.02$.

On this way there are some changes of the voltages and the currents of the circuit but they could be considered negligible, so it's considered true the past result from now too.

With $\lambda = 0.02$, the r_0 has a finite value:

$$r_0 = \frac{1}{\lambda I_{S_A}} = \frac{1}{0.02 \cdot 150.06433\mu A} \simeq 333.2k\Omega \quad (1.57)$$

It's possible see the r_0 resitances in the small signal circuit by using the PI model of the transistor MOSFET (figure 1.3).

In the small signal circuit are considered only the alternate sources, so the V_{DD} and the V_{SS} voltage sources are substituted by a short circuit.

In this way the mirror current source composed by the MOSFET M_4 and the MOSFET M_5 are off because the drain, the source and the gate of these MOSFETs have no voltage applied. So we can simplify the small signal circuit of the figure 1.3 to the small signal circuit of the figure 1.4.

Now it could be clear the effect of the $\lambda = 0.02$ of the MOSFET M_3 because its resistance r_0 has an effect to the gain of the circuit if the voltage of the node S_A is different than null.


1.3.2 Differential pure signal gain

Now it's applied small alternate voltage signal to the circuit of the figure 1.4:

$$V_{i_1} = +\frac{V_{id}}{2} \quad (1.58)$$

$$V_{i_2} = -\frac{V_{id}}{2} \quad (1.59)$$

Applying a small alternate voltage signal to the circuit of the figure 1.4 the resistance r_0 could be treated as a short circuit because the current on it is null.



--- Z:\home\peter\Git\PPI\Electronic\SPICE\Exercises\1819\ThirdExercise\schematics\IDifferentialAmplifierSmallSignalPiComplete.asc ---

Figure 1.3: Small signal circuit by using the PI model of the Transistor MOSFET

Given that the circuit of the figure 1.4 is symmetrical and it's possible analyse the single ended behaviour of the circuit on figure 1.6.

It's possible finding the single ended gain A_d for a small differential signal by calculate first the single ended gain (eq. 1.60, 1.61 and 1.67).

$$V_{D1} = -g_m V_{G1} R_{D1} \xrightarrow{(eq.1.58)} V_{D2} = -g_m \frac{V_{id}}{2} R_{D2} \implies \frac{V_{D1}}{V_{id}} = -\frac{g_m R_{D1}}{2} = A_{d_{single-ended.1}} \quad (1.60)$$

$$V_{D2} = -g_m V_{G2} R_{D2} \xrightarrow{(eq.1.59)} V_{D2} = -g_m \left(-\frac{V_{id}}{2}\right) R_{D2} \implies \frac{V_{D2}}{V_{id}} = \frac{g_m R_{D2}}{2} = A_{d_{single-ended.2}} \quad (1.61)$$

$$A_d = \frac{V_{D2} - V_{D1}}{V_{id}} \quad (1.62)$$

$$= A_{d_{single-ended.2}} - A_{d_{single-ended.1}} \quad (1.63)$$

$$= \frac{g_m R_{D2}}{2} - \left(-\frac{g_m R_{D1}}{2}\right) \quad (1.64)$$

$$= \frac{g_m R_{D2}}{2} + \frac{g_m R_{D1}}{2} \quad (1.65)$$

$$= \frac{g_m R_D}{2} + \frac{g_m R_D}{2} \quad (\text{see eq. 1.6}) \quad (1.66)$$

$$= g_m R_D \quad (1.67)$$

1.3.3 Common mode gain


By applying a common mode signal to the inputs of the differential amplifier, on r_0 there is current different than null, so in this case it gives an important role to the common mode gain.

Some useful equations for calculating the common mode gain are 1.69 and 1.74.

$$V_{i1} = V_{i2} = V_{i_{CM}} \quad (1.68)$$

$$R_{ss1} = R_{ss2} = 2r_0 \quad (1.69)$$

$$V_{G1SA} = V_{G2SA} \quad (1.70)$$



--- Z:\home\peter\Git\PPI\ElectronicSPICEExercises1819\ThirdExercise\schematics\DifferentialAmplifierSmallSignalPi.asc ---

Figure 1.4: Small signal circuit by using the PI model of the Transistor MOSFET - simplifications

$$-V_{i1} + V_{G1SA} + g_m V_{G1SA} R_{ss1} = 0 \quad (1.71)$$

$$-V_{i1} + V_{G1SA}(1 + g_m R_{ss1}) = 0 \quad (1.72)$$

$$\Rightarrow V_{G1SA} = \frac{V_{i1}}{1 + g_m R_{ss1}} \quad (1.73)$$

$$\xrightarrow{(eq.:1.69)} V_{G1SA} = \frac{V_{i1}}{1 + 2g_m r_0} \quad (1.74)$$

In order to calculate the common mode gain, the common mode gain of the single ended circuit are calculated first:

$$V_{D1} = -g_m R_{D1} V_{G1SA} \quad (1.75)$$

$$V_{D1} = -g_m R_{D1} \left(\frac{V_{i1}}{1 + 2g_m r_0} \right) \quad (1.76)$$

$$V_{D1} = \frac{-g_m R_{D1}}{1 + 2g_m r_0} V_{i1} \quad (1.77)$$

$$\Rightarrow \frac{V_{D1}}{V_{i1}} = \frac{-g_m R_{D1}}{1 + 2g_m r_0} \quad (1.78)$$

$$\xrightarrow{(eq.:1.6 \text{ and } 1.68)} \frac{V_{D1}}{V_{iCM}} = \frac{-g_m R_D}{1 + 2g_m r_0} = A_{CM_{single-ended.1}} \simeq \frac{R_D}{2r_0} \quad (1.79)$$


$$A_{CM_{single-ended.1}} \simeq \frac{R_D}{2r_0} \quad \text{for } r_0 \gg 1 \quad (1.80)$$

$$\frac{V_{D2}}{V_{iCM}} = \frac{-g_m R_D}{1 + 2g_m r_0} = A_{CM_{single-ended.2}} \quad (1.81)$$

$$(1.82)$$

$$A_{CM_{single-ended.2}} \simeq \frac{R_D}{2r_0} \quad \text{for } r_0 \gg 1 \quad (1.83)$$

The common mode gain is:



--- Z:\homelpeter\Git\PPI\ElectronicSPICEExercises\1819\ThirdExercise\schematics\DifferentialAmplifierSmallSignalPiSingleEndedVID.asc ---

Figure 1.5: Small signal circuit single ended for a differential pure signal.

$$A_{CM} = \frac{V_{D2} - V_{D1}}{V_{i_{CM}}} \quad (1.84)$$

$$= A_{CM_{single-ended.2}} - A_{CM_{single-ended.1}} \quad (1.85)$$


$$= \left(\frac{-g_m R_D}{1 + 2g_m r_0} \right) - \left(\frac{-g_m R_D}{1 + 2g_m r_0} \right) \quad (1.86)$$

$$= 0 \quad (1.87)$$

1.3.4 Common mode gain with $\lambda_3 = 0$

With $\lambda_3 = 0$, the r_0 of the MOSFET M_3 isn't finite.

So the common mode gain would be null but also the single ended gain.



--- Z:\homelpeter\Git\PElectronicSPICEExercises1819\ThirdExercises\schematics\DifferentialAmplifierSmallSignalPISingleEndedCM.asc ---

Figure 1.6: Small signal circuit single ended for a common mode signal.

1.4 SPICE analysis

1.4.1 $R_D = 20k$

Operating Point on static conditions ($\lambda_3 = 0$)

```
* Differential Amplifier – Static conditions
*****
* 3st Exercise – Fundamentals Of Electronics – a.a. 2018–2019 – UniPD – Italy *
*                               Pietro Prandini – mat. 1097752                               *
*                                                                                               *
* This work is licensed under the Creative Commons Attribution–ShareAlike 4.0 *
* International License. To view a copy of this license, visit                               *
* http://creativecommons.org/licenses/by-sa/4.0/ or send a letter to Creative *
* Commons, PO Box 1866, Mountain View, CA 94042, USA. *
*****

* Parameters
.param RD = 20k

* NMOS models
.model NA NMOS VT0=0.5 KP=200u LAMBDA=0 W=4.00u L=0.20u
.model NS NMOS VT0=0.5 KP=200u LAMBDA=0 W=1.25u L=0.25u

* Resistances
RD1 DD D1 {RD}
RD2 DD D2 {RD}
RD4 DD D4 33k

* Transistors
M1 D1 G1 SA SA NA
M2 D2 G2 SA SA NA
M3 SA D4 SS SS NS
M4 D4 D4 SS SS NS

* Voltage sources
VDD DD 0 3
VSS SS 0 -3
Vi1 G1 0 0
Vi2 G2 0 0

* Analysis
.op

.END
```

—— Operating Point ——

V(dd):	3	voltage
V(d1):	1.49935	voltage
V(d2):	1.49935	voltage
V(d4):	-1.95216	voltage
V(g1):	0	voltage
V(sa):	-0.693691	voltage
V(g2):	0	voltage
V(ss):	-3	voltage
Id(M4):	0.000150065	device_current
Ig(M4):	0	device_current
Ib(M4):	-1.05784e-012	device_current
Is(M4):	-0.000150065	device_current
Id(M3):	0.000150065	device_current

Ig(M3):	0	device_current
Ib(M3):	-2.31631e-012	device_current
Is(M3):	-0.000150065	device_current
Id(M2):	7.50327e-005	device_current
Ig(M2):	0	device_current
Ib(M2):	-2.20304e-012	device_current
Is(M2):	-7.50327e-005	device_current
Id(M1):	7.50327e-005	device_current
Ig(M1):	0	device_current
Ib(M1):	-2.20304e-012	device_current
Is(M1):	-7.50327e-005	device_current
I(Rd4):	0.000150065	device_current
I(Rd2):	7.50327e-005	device_current
I(Rd1):	7.50327e-005	device_current
I(Vi2):	0	device_current
I(Vi1):	0	device_current
I(Vss):	0.000300131	device_current
I(Vdd):	-0.000300131	device_current

Operating Point - common mode signals

```

* Differential Amplifier – Common Mode signals
*****
* 3st Exercise – Fundamentals Of Electronics – a.a. 2018–2019 – UniPD – Italy *
*                               Pietro Prandini – mat. 1097752                *
*                                                                           *
* This work is licensed under the Creative Commons Attribution–ShareAlike 4.0 *
* International License. To view a copy of this license, visit              *
* http://creativecommons.org/licenses/by-sa/4.0/ or send a letter to Creative *
* Commons, PO Box 1866, Mountain View, CA 94042, USA.                      *
*****

* Parameters
.param RD = 20k

* NMOS models
.model NA NMOS VT0=0.5 KP=200u LAMBDA=0 W=4.00u L=0.20u
.model NS NMOS VT0=0.5 KP=200u LAMBDA=0 W=1.25u L=0.25u
.model NS3 NMOS VT0=0.5 KP=200u LAMBDA=0.02 W=1.25u L=0.25u

* Resistances
RD1 DD D1 {RD}
RD2 DD D2 {RD}
RD4 DD D4 33k

* Transistors
M1 D1 G1 SA SA NA
M2 D2 G2 SA SA NA
M3 SA D4 SS SS NS3
M4 D4 D4 SS SS NS

* Voltage sources
VDD DD 0 3
VSS SS 0 -3
Vi1 G1 0 sine(0 10m 10k 0 0 0)
Vi2 G2 0 sine(0 10m 10k 0 0 0)

* Analysis
.op

```

.END

— Operating Point —

V(dd):	3	voltage
V(d1):	1.43026	voltage
V(d2):	1.43026	voltage
V(d4):	-1.95216	voltage
V(g1):	0	voltage
V(sa):	-0.698101	voltage
V(g2):	0	voltage
V(ss):	-3	voltage
Id(M4):	0.000150065	device_current
Ig(M4):	0	device_current
Ib(M4):	-1.05784e-012	device_current
Is(M4):	-0.000150065	device_current
Id(M3):	0.000156974	device_current
Ig(M3):	0	device_current
Ib(M3):	-2.3119e-012	device_current
Is(M3):	-0.000156974	device_current
Id(M2):	7.84879e-005	device_current
Ig(M2):	0	device_current
Ib(M2):	-2.13836e-012	device_current
Is(M2):	-7.84879e-005	device_current
Id(M1):	7.84879e-005	device_current
Ig(M1):	0	device_current
Ib(M1):	-2.13836e-012	device_current
Is(M1):	-7.84879e-005	device_current
I(Rd4):	0.000150065	device_current
I(Rd2):	7.8487e-005	device_current
I(Rd1):	7.8487e-005	device_current
I(Vi2):	0	device_current
I(Vi1):	0	device_current
I(Vss):	0.00030704	device_current
I(Vdd):	-0.00030704	device_current

Operating Point - differential signals

```

* Differential Amplifier – Differential signals
*****
* 3st Exercise – Fundamentals Of Electronics – a.a. 2018–2019 – UniPD – Italy *
*                               Pietro Prandini – mat. 1097752                      *
*                                                                                     *
* This work is licensed under the Creative Commons Attribution–ShareAlike 4.0 *
* International License. To view a copy of this license, visit                      *
* http://creativecommons.org/licenses/by-sa/4.0/ or send a letter to Creative *
* Commons, PO Box 1866, Mountain View, CA 94042, USA.                             *
*****

* Parameters
.param RD = 20k

* NMOS models
.model NA NMOS VT0=0.5 KP=200u LAMBDA=0 W=4.00u L=0.20u
.model NS NMOS VT0=0.5 KP=200u LAMBDA=0 W=1.25u L=0.25u
.model NS3 NMOS VT0=0.5 KP=200u LAMBDA=0.02 W=1.25u L=0.25u

* Resistances
RD1 DD D1 {RD}
RD2 DD D2 {RD}

```



```

RD4 DD D4 33k

* Transistors
M1 D1 G1 SA SA NA
M2 D2 G2 SA SA NA
M3 SA D4 SS SS NS3
M4 D4 D4 SS SS NS

* Voltage sources
VDD DD 0 3
VSS SS 0 -3
Vi1 G1 0 sine(0 10m 10k 0 0 0)
Vi2 0 G2 sine(0 10m 10k 0 0 0)

* Analysis
.op

.END

```

—— Operating Point ——

V(dd):	3	voltage
V(d1):	1.49935	voltage
V(d2):	1.49935	voltage
V(d4):	-1.95216	voltage
V(g1):	0	voltage
V(sa):	-0.693691	voltage
V(g2):	0	voltage
V(ss):	-3	voltage
Id(M4):	0.000150065	device_current
Ig(M4):	0	device_current
Ib(M4):	-1.05784e-012	device_current
Is(M4):	-0.000150065	device_current
Id(M3):	0.000150065	device_current
Ig(M3):	0	device_current
Ib(M3):	-2.31631e-012	device_current
Is(M3):	-0.000150065	device_current
Id(M2):	7.50327e-005	device_current
Ig(M2):	0	device_current
Ib(M2):	-2.20304e-012	device_current
Is(M2):	-7.50327e-005	device_current
Id(M1):	7.50327e-005	device_current
Ig(M1):	0	device_current
Ib(M1):	-2.20304e-012	device_current
Is(M1):	-7.50327e-005	device_current
I(Rd4):	0.000150065	device_current
I(Rd2):	7.50327e-005	device_current
I(Rd1):	7.50327e-005	device_current
I(Vi2):	0	device_current
I(Vi1):	0	device_current
I(Vss):	0.000300131	device_current
I(Vdd):	-0.000300131	device_current

1.4.2 $R_D = 1097752/100$

Operating Point on static conditions ($\lambda_3 = 0$)

```

* Differential Amplifier - Static conditions
*****
* 3st Exercise - Fundamentals Of Electronics - a.a. 2018-2019 - UniPD - Italy *

```

```

*                               Pietro Prandini – mat. 1097752                               *
*                                                                                             *
* This work is licensed under the Creative Commons Attribution–ShareAlike 4.0 *
* International License. To view a copy of this license, visit *
* http://creativecommons.org/licenses/by-sa/4.0/ or send a letter to Creative *
* Commons, PO Box 1866, Mountain View, CA 94042, USA. *
*                                                                                             *
*****

* Parameters
.param RD = {1097752/100}

* NMOS models
.model NA NMOS VT0=0.5 KP=200u LAMBDA=0 W=4.00u L=0.20u
.model NS NMOS VT0=0.5 KP=200u LAMBDA=0 W=1.25u L=0.25u

* Resistances
RD1 DD D1 {RD}
RD2 DD D2 {RD}
RD4 DD D4 33k

* Transistors
M1 D1 G1 SA SA NA
M2 D2 G2 SA SA NA
M3 SA D4 SS SS NS
M4 D4 D4 SS SS NS

* Voltage sources
VDD DD 0 3
VSS SS 0 -3
Vi1 G1 0 0
Vi2 G2 0 0

* Analysis
.op

.END

```

—— Operating Point ——

V(dd):	3	voltage
V(d1):	2.17633	voltage
V(d2):	2.17633	voltage
V(d4):	-1.95216	voltage
V(g1):	0	voltage
V(sa):	-0.693691	voltage
V(g2):	0	voltage
V(ss):	-3	voltage
Id(M4):	0.000150065	device_current
Ig(M4):	0	device_current
Ib(M4):	-1.05784e-012	device_current
Is(M4):	-0.000150065	device_current
Id(M3):	0.000150065	device_current
Ig(M3):	0	device_current
Ib(M3):	-2.31631e-012	device_current
Is(M3):	-0.000150065	device_current
Id(M2):	7.50327e-005	device_current
Ig(M2):	0	device_current
Ib(M2):	-2.88002e-012	device_current
Is(M2):	-7.50327e-005	device_current
Id(M1):	7.50327e-005	device_current
Ig(M1):	0	device_current

Ib(M1):	-2.88002e-012	device_current
Is(M1):	-7.50327e-005	device_current
I(Rd4):	0.000150065	device_current
I(Rd2):	7.50327e-005	device_current
I(Rd1):	7.50327e-005	device_current
I(Vi2):	0	device_current
I(Vi1):	0	device_current
I(Vss):	0.000300131	device_current
I(Vdd):	-0.000300131	device_current

Operating Point - common mode signals

```

* Differential Amplifier – Common Mode signals
*****
* 3st Exercise – Fundamentals Of Electronics – a.a. 2018–2019 – UniPD – Italy *
*                               Pietro Prandini – mat. 1097752                *
*                                                                           *
* This work is licensed under the Creative Commons Attribution–ShareAlike 4.0 *
* International License. To view a copy of this license, visit              *
* http://creativecommons.org/licenses/by-sa/4.0/ or send a letter to Creative *
* Commons, PO Box 1866, Mountain View, CA 94042, USA.                      *
*****

* Parameters
.param RD = {1097752/100}

* NMOS models
.model NA NMOS VT0=0.5 KP=200u LAMBDA=0 W=4.00u L=0.20u
.model NS NMOS VT0=0.5 KP=200u LAMBDA=0 W=1.25u L=0.25u
.model NS3 NMOS VT0=0.5 KP=200u LAMBDA=0.02 W=1.25u L=0.25u

* Resistances
RD1 DD D1 {RD}
RD2 DD D2 {RD}
RD4 DD D4 33k

* Transistors
M1 D1 G1 SA SA NA
M2 D2 G2 SA SA NA
M3 SA D4 SS SS NS3
M4 D4 D4 SS SS NS

* Voltage sources
VDD DD 0 3
VSS SS 0 -3
Vi1 G1 0 sine(0 10m 10k 0 0 0)
Vi2 G2 0 sine(0 10m 10k 0 0 0)

* Analysis
.op

.END

```

— Operating Point —

V(dd):	3	voltage
V(d1):	2.13841	voltage
V(d2):	2.13841	voltage
V(d4):	-1.95216	voltage
V(g1):	0	voltage

V(sa):	-0.698101	voltage
V(g2):	0	voltage
V(ss):	-3	voltage
Id(M4):	0.000150065	device_current
Ig(M4):	0	device_current
Ib(M4):	-1.05784e-012	device_current
Is(M4):	-0.000150065	device_current
Id(M3):	0.000156974	device_current
Ig(M3):	0	device_current
Ib(M3):	-2.3119e-012	device_current
Is(M3):	-0.000156974	device_current
Id(M2):	7.84879e-005	device_current
Ig(M2):	0	device_current
Ib(M2):	-2.84651e-012	device_current
Is(M2):	-7.84879e-005	device_current
Id(M1):	7.84879e-005	device_current
Ig(M1):	0	device_current
Ib(M1):	-2.84651e-012	device_current
Is(M1):	-7.84879e-005	device_current
I(Rd4):	0.000150065	device_current
I(Rd2):	7.8487e-005	device_current
I(Rd1):	7.8487e-005	device_current
I(Vi2):	0	device_current
I(Vi1):	0	device_current
I(Vss):	0.00030704	device_current
I(Vdd):	-0.00030704	device_current

Operating Point - differential signals

```

* Differential Amplifier – Differential signals
*****
* 3st Exercise – Fundamentals Of Electronics – a.a. 2018–2019 – UniPD – Italy *
*                               Pietro Prandini – mat. 1097752 *
*
* This work is licensed under the Creative Commons Attribution–ShareAlike 4.0 *
* International License. To view a copy of this license, visit *
* http://creativecommons.org/licenses/by-sa/4.0/ or send a letter to Creative *
* Commons, PO Box 1866, Mountain View, CA 94042, USA. *
*****

* Parameters
.param RD = {1097752/100}

* NMOS models
.model NA NMOS VT0=0.5 KP=200u LAMBDA=0 W=4.00u L=0.20u
.model NS NMOS VT0=0.5 KP=200u LAMBDA=0 W=1.25u L=0.25u
.model NS3 NMOS VT0=0.5 KP=200u LAMBDA=0.02 W=1.25u L=0.25u

* Resistances
RD1 DD D1 {RD}
RD2 DD D2 {RD}
RD4 DD D4 33k

* Transistors
M1 D1 G1 SA SA NA
M2 D2 G2 SA SA NA
M3 SA D4 SS SS NS3
M4 D4 D4 SS SS NS

* Voltage sources

```

```

VDD DD 0 3
VSS SS 0 -3
Vi1 G1 0 sine(0 10m 10k 0 0 0)
Vi2 0 G2 sine(0 10m 10k 0 0 0)

* Analysis
.op

.END

```

—— Operating Point ——

```

V(dd):      3          voltage
V(d1):      2.13841    voltage
V(d2):      2.13841    voltage
V(d4):      -1.95216    voltage
V(g1):      0          voltage
V(sa):      -0.698101  voltage
V(g2):      0          voltage
V(ss):      -3         voltage
Id(M4):      0.000150065 device_current
Ig(M4):      0          device_current
Ib(M4):      -1.05784e-012 device_current
Is(M4):      -0.000150065 device_current
Id(M3):      0.000156974 device_current
Ig(M3):      0          device_current
Ib(M3):      -2.3119e-012 device_current
Is(M3):      -0.000156974 device_current
Id(M2):      7.84879e-005 device_current
Ig(M2):      0          device_current
Ib(M2):      -2.84651e-012 device_current
Is(M2):      -7.84879e-005 device_current
Id(M1):      7.84879e-005 device_current
Ig(M1):      0          device_current
Ib(M1):      -2.84651e-012 device_current
Is(M1):      -7.84879e-005 device_current
I(Rd4):      0.000150065 device_current
I(Rd2):      7.8487e-005 device_current
I(Rd1):      7.8487e-005 device_current
I(Vi2):      0          device_current
I(Vi1):      0          device_current
I(Vss):      0.00030704 device_current
I(Vdd):      -0.00030704 device_current

```

1.4.3 $R_D = 1097752/1000$

Operating Point on static conditions ($\lambda_3 = 0$)

```

* Differential Amplifier – Static conditions
*****
* 3st Exercise – Fundamentals Of Electronics – a.a. 2018–2019 – UniPD – Italy *
*                               Pietro Prandini – mat. 1097752                      *
*                                                                                     *
* This work is licensed under the Creative Commons Attribution–ShareAlike 4.0 *
* International License. To view a copy of this license, visit                  *
* http://creativecommons.org/licenses/by-sa/4.0/ or send a letter to Creative *
* Commons, PO Box 1866, Mountain View, CA 94042, USA.                          *
*****
* Parameters

```

```

.param RD = {1097752/1000}

* NMOS models
.model NA NMOS VT0=0.5 KP=200u LAMBDA=0 W=4.00u L=0.20u
.model NS NMOS VT0=0.5 KP=200u LAMBDA=0 W=1.25u L=0.25u

* Resistances
RD1 DD D1 {RD}
RD2 DD D2 {RD}
RD4 DD D4 33k

* Transistors
M1 D1 G1 SA SA NA
M2 D2 G2 SA SA NA
M3 SA D4 SS SS NS
M4 D4 D4 SS SS NS

* Voltage sources
VDD DD 0 3
VSS SS 0 -3
Vi1 G1 0 0
Vi2 G2 0 0

* Analysis
.op

.END

```

—— Operating Point ——

V(dd):	3	voltage
V(d1):	2.91763	voltage
V(d2):	2.91763	voltage
V(d4):	-1.95216	voltage
V(g1):	0	voltage
V(sa):	-0.693691	voltage
V(g2):	0	voltage
V(ss):	-3	voltage
Id(M4):	0.000150065	device_current
Ig(M4):	0	device_current
Ib(M4):	-1.05784e-012	device_current
Is(M4):	-0.000150065	device_current
Id(M3):	0.000150065	device_current
Ig(M3):	0	device_current
Ib(M3):	-2.31631e-012	device_current
Is(M3):	-0.000150065	device_current
Id(M2):	7.50327e-005	device_current
Ig(M2):	0	device_current
Ib(M2):	-3.62132e-012	device_current
Is(M2):	-7.50327e-005	device_current
Id(M1):	7.50327e-005	device_current
Ig(M1):	0	device_current
Ib(M1):	-3.62132e-012	device_current
Is(M1):	-7.50327e-005	device_current
I(Rd4):	0.000150065	device_current
I(Rd2):	7.50327e-005	device_current
I(Rd1):	7.50327e-005	device_current
I(Vi2):	0	device_current
I(Vi1):	0	device_current
I(Vss):	0.000300131	device_current
I(Vdd):	-0.000300131	device_current

Operating Point - common mode signals

```

* Differential Amplifier – Common Mode signals
*****
* 3st Exercise – Fundamentals Of Electronics – a.a. 2018–2019 – UniPD – Italy *
*                               Pietro Prandini – mat. 1097752                *
*                                                                           *
* This work is licensed under the Creative Commons Attribution–ShareAlike 4.0 *
* International License. To view a copy of this license, visit              *
* http://creativecommons.org/licenses/by-sa/4.0/ or send a letter to Creative *
* Commons, PO Box 1866, Mountain View, CA 94042, USA.                      *
*****

* Parameters
.param RD = {1097752/1000}

* NMOS models
.model NA NMOS VT0=0.5 KP=200u LAMBDA=0 W=4.00u L=0.20u
.model NS NMOS VT0=0.5 KP=200u LAMBDA=0 W=1.25u L=0.25u
.model NS3 NMOS VT0=0.5 KP=200u LAMBDA=0.02 W=1.25u L=0.25u

* Resistances
RD1 DD D1 {RD}
RD2 DD D2 {RD}
RD4 DD D4 33k

* Transistors
M1 D1 G1 SA SA NA
M2 D2 G2 SA SA NA
M3 SA D4 SS SS NS3
M4 D4 D4 SS SS NS

* Voltage sources
VDD DD 0 3
VSS SS 0 -3
Vi1 G1 0 sine(0 10m 10k 0 0 0)
Vi2 G2 0 sine(0 10m 10k 0 0 0)

* Analysis
.op

.END

```

— Operating Point —

V(dd):	3	voltage
V(d1):	2.91384	voltage
V(d2):	2.91384	voltage
V(d4):	-1.95216	voltage
V(g1):	0	voltage
V(sa):	-0.698101	voltage
V(g2):	0	voltage
V(ss):	-3	voltage
Id(M4):	0.000150065	device_current
Ig(M4):	0	device_current
Ib(M4):	-1.05784e-012	device_current
Is(M4):	-0.000150065	device_current
Id(M3):	0.000156974	device_current

Ig(M3):	0	device_current
Ib(M3):	-2.3119e-012	device_current
Is(M3):	-0.000156974	device_current
Id(M2):	7.84879e-005	device_current
Ig(M2):	0	device_current
Ib(M2):	-3.62194e-012	device_current
Is(M2):	-7.84879e-005	device_current
Id(M1):	7.84879e-005	device_current
Ig(M1):	0	device_current
Ib(M1):	-3.62194e-012	device_current
Is(M1):	-7.84879e-005	device_current
I(Rd4):	0.000150065	device_current
I(Rd2):	7.8487e-005	device_current
I(Rd1):	7.8487e-005	device_current
I(Vi2):	0	device_current
I(Vi1):	0	device_current
I(Vss):	0.00030704	device_current
I(Vdd):	-0.00030704	device_current

Operating Point - differential signals

```

* Differential Amplifier – Differential signals
*****
* 3st Exercise – Fundamentals Of Electronics – a.a. 2018–2019 – UniPD – Italy *
*                               Pietro Prandini – mat. 1097752                      *
*                                                                                   *
* This work is licensed under the Creative Commons Attribution–ShareAlike 4.0 *
* International License. To view a copy of this license, visit                  *
* http://creativecommons.org/licenses/by-sa/4.0/ or send a letter to Creative *
* Commons, PO Box 1866, Mountain View, CA 94042, USA.                          *
*****

* Parameters
.param RD = {1097752/1000}

* NMOS models
.model NA NMOS VT0=0.5 KP=200u LAMBDA=0 W=4.00u L=0.20u
.model NS NMOS VT0=0.5 KP=200u LAMBDA=0 W=1.25u L=0.25u
.model NS3 NMOS VT0=0.5 KP=200u LAMBDA=0.02 W=1.25u L=0.25u

* Resistances
RD1 DD D1 {RD}
RD2 DD D2 {RD}
RD4 DD D4 33k

* Transistors
M1 D1 G1 SA SA NA
M2 D2 G2 SA SA NA
M3 SA D4 SS SS NS3
M4 D4 D4 SS SS NS

* Voltage sources
VDD DD 0 3
VSS SS 0 -3
Vi1 G1 0 sine(0 10m 10k 0 0 0)
Vi2 0 G2 sine(0 10m 10k 0 0 0)

* Analysis
.op

```


.END

—— Operating Point ——

V(dd):	3	voltage
V(d1):	2.91384	voltage
V(d2):	2.91384	voltage
V(d4):	-1.95216	voltage
V(g1):	0	voltage
V(sa):	-0.698101	voltage
V(g2):	0	voltage
V(ss):	-3	voltage
Id(M4):	0.000150065	device_current
Ig(M4):	0	device_current
Ib(M4):	-1.05784e-012	device_current
Is(M4):	-0.000150065	device_current
Id(M3):	0.000156974	device_current
Ig(M3):	0	device_current
Ib(M3):	-2.3119e-012	device_current
Is(M3):	-0.000156974	device_current
Id(M2):	7.84879e-005	device_current
Ig(M2):	0	device_current
Ib(M2):	-3.62194e-012	device_current
Is(M2):	-7.84879e-005	device_current
Id(M1):	7.84879e-005	device_current
Ig(M1):	0	device_current
Ib(M1):	-3.62194e-012	device_current
Is(M1):	-7.84879e-005	device_current
I(Rd4):	0.000150065	device_current
I(Rd2):	7.8487e-005	device_current
I(Rd1):	7.8487e-005	device_current
I(Vi2):	0	device_current
I(Vi1):	0	device_current
I(Vss):	0.00030704	device_current
I(Vdd):	-0.00030704	device_current

Chapter 2

CMOS Inverter




Figure 2.1: CMOS Inverter

2.1 Initial data

$$V_{DD} = 5V \quad (2.1)$$

$$M_p \begin{cases} V_{t_p} = -0.7V \\ K'_p = \mu_p C_{ox} = 12\mu A/V^2 \\ \lambda = 0.01V^{-1} \end{cases} \quad (2.2)$$

$$M_n \begin{cases} V_{t_n} = 0.7V \\ K'_n = \mu_n C_{ox} = 48\mu A/V^2 \\ \lambda = 0.01V^{-1} \end{cases} \quad (2.3)$$

2.2 Dimensioning a balanced inverter

The conditions for a balanced inverter is on equation 2.4.

It's granted that the transfer function of the inverter is symmetrical.

$$\frac{W_p}{W_n} = \frac{\mu_n}{\mu_p} \quad (2.4)$$

The length of the channel is usually setted to the minimum length of the fabrication process. In this case it's supposed to be as in the equation 2.5.

$$L_p = L_n = 50nm \quad (2.5)$$

Furthermore, in order to minimizing the area of the integrated circuit, the weight-length's rapport of the NMOS is usually setted between 1 and 1.5.

In this case it's supposed to be as in the equation 2.6.

$$\frac{W_n}{L_n} = 1 \quad (2.6)$$

So in the equations 2.7 and 2.8 there is the weight of the MOSFET's circuit.

$$\frac{W_n}{L_n} = 1 \implies W_n = L_n \xrightarrow{(eq.2.5)} W_n = 50nm \quad (2.7)$$

$$\frac{W_p}{W_n} = \frac{\mu_n}{\mu_p} \implies W_p = \frac{\mu_p}{\mu_n} W_n \xrightarrow{(eq.2.3,2.2,2.7)} W_p = \frac{48\mu A/V^2}{12\mu A/V^2} \cdot 50nm = 200nm \quad (2.8)$$

2.2.1 Resuming

MOS	$K'_{p,n}$	$W_{p,n}$	$L_{p,n}$	$V_{t_{p,n}}$	λ
M_p	$12\mu A/V^2$	$200nm$	$50nm$	$-0.7V$	$0.01V^{-1}$
M_n	$48\mu A/V^2$	$50nm$	$50nm$	$0.7V$	$0.01V^{-1}$

2.3 Characteristic V_{IN}/V_{OUT}

```

* CMOS Inverter – Characteristic Vin/Vout
*****
* 3st Exercise – Fundamentals Of Electronics – a.a. 2018–2019 – UniPD – Italy *
*                               Pietro Prandini – mat. 1097752 *
*
* This work is licensed under the Creative Commons Attribution–ShareAlike 4.0 *
* International License. To view a copy of this license, visit *
* http://creativecommons.org/licenses/by-sa/4.0/ or send a letter to Creative *
* Commons, PO Box 1866, Mountain View, CA 94042, USA. *
*****

* NMOS models
.model P PMOS VT0=-0.7 KP=12u LAMBDA=0.01 L=50n W=200n
.model N NMOS VT0=0.7 KP=48u LAMBDA=0.01 L=50n W=50n

* Transistors
Mp OUT IN DD DD P
Mn OUT IN 0 0 N

* Voltage sources
VDD DD 0 5
VIN IN 0 0

* Analysis
.DC VIN 0 5

.END

```

The output could be visible on figure 2.2.




Figure 2.2: CMOS Inverter - V_{OUT}/V_{IN} characteristic.

2.4 Applying a square wave on input

```
* CMOS Inverter - Square wave
*****
* 3st Exercise - Fundamentals Of Electronics - a.a. 2018-2019 - UniPD - Italy *
*                               Pietro Prandini - mat. 1097752                *
*                                                                           *
* This work is licensed under the Creative Commons Attribution-ShareAlike 4.0 *
* International License. To view a copy of this license, visit              *
* http://creativecommons.org/licenses/by-sa/4.0/ or send a letter to Creative *
* Commons, PO Box 1866, Mountain View, CA 94042, USA.                       *
*****

* NMOS models
.model P PMOS VT0=-0.7 KP=12u LAMBDA=0.01 L=50n W=200n
.model N NMOS VT0=0.7 KP=48u LAMBDA=0.01 L=50n W=50n

* Transistors
Mp OUT IN DD DD P
Mn OUT IN 0 0 N

* Voltage sources
VDD DD 0 5
VIN IN 0 pulse(0V 5V 0 0 0 {1/(2MEG)} {1/(1MEG)})

* Analysis
.TRAN 0 2.5u 0

.END
```

The output could be visible on figure 2.3.




Figure 2.3: CMOS Inverter - A square wave on input.

2.5 Characteristic V_{IN}/V_{OUT} - Changing W_n

```

* CMOS Inverter - Characteristic Vin/Vout
*****
* 3st Exercise - Fundamentals Of Electronics - a.a. 2018-2019 - UniPD - Italy *
*                               Pietro Prandini - mat. 1097752                *
*                                                                           *
* This work is licensed under the Creative Commons Attribution-ShareAlike 4.0 *
* International License. To view a copy of this license, visit              *
* http://creativecommons.org/licenses/by-sa/4.0/ or send a letter to Creative *
* Commons, PO Box 1866, Mountain View, CA 94042, USA.                      *
*****

* NMOS models
.model P PMOS VT0=-0.7 KP=12u LAMBDA=0.01 L=50n W=200n
.model N NMOS VT0=0.7 KP=48u LAMBDA=0.01 L={50n*{Ln}} W={50n*{Wn}}

* Transistors
Mp OUT IN DD DD P
Mn OUT IN 0 0 N

* Voltage sources
VDD DD 0 5
VIN IN 0 0


* Parameters
.param Ln = 10
.step param Wn list 1 2 5 20 50 100

* Analysis
.DC VIN 0 5

.END

```

The output could be visible on figure 2.4.

Figure 2.4: CMOS Inverter - V_{OUT}/V_{IN} characteristic varying W_n .

2.6 Applying a square wave on input - Changing W_n

```

* CMOS Inverter - Square wave
*****
* 3st Exercise - Fundamentals Of Electronics - a.a. 2018-2019 - UniPD - Italy *
*                               Pietro Prandini - mat. 1097752                *
*                                                                           *
* This work is licensed under the Creative Commons Attribution-ShareAlike 4.0 *
* International License. To view a copy of this license, visit              *
* http://creativecommons.org/licenses/by-sa/4.0/ or send a letter to Creative *
* Commons, PO Box 1866, Mountain View, CA 94042, USA.                      *
*****

* NMOS models
.model P PMOS VT0=-0.7 KP=12u LAMBDA=0.01 L=50n W=200n
.model N NMOS VT0=0.7 KP=48u LAMBDA=0.01 L={50n*{Ln}} W={50n*{Wn}}

* Transistors
Mp OUT IN DD DD P
Mn OUT IN 0 0 N

* Voltage sources
VDD DD 0 5
VIN IN 0 pulse(0V 5V 0 0 0 {1/(2MEG)} {1/(1MEG)})

* Parameters
.param Ln = 10
.step param Wn list 1 2 5 20 50 100

* Analysis
.TRAN 0 2.5u 0

.END

```

The output could be visible on figure 2.5.




Figure 2.5: CMOS Inverter - A square wave on input varying W_n .